
Code for light following robot follows.  Be sure that you make sure all the comments stay comments if 
you copy-paste:

#include <UsbSimple.h>//You can use Serial.h if you are using a regular arduino

int state=10;//Set state to something, to ensure it will change

//Set up GPIO pins
void setup(){
  USB.begin();       //Start usb conenction
  
  pinMode(9,OUTPUT); //Right engine forward
  pinMode(10,OUTPUT);//Left engine forward
  
  pinMode(7,OUTPUT);//Right engine forward indicator LED
  pinMode(8,OUTPUT);//Left engine forward indicator LED
  
  pinMode(A0,INPUT);//right sensor
  pinMode(A1,INPUT);//left sensor

  digitalWrite(A0, HIGH);//turn on internal pull up resistors
  digitalWrite(A1,HIGH);
  
}

/*Main loop:
This loops through two functions, checkSensors and roboGo.
checkSensors checks sensors and returns 1,2,3, or 0.  1-3 means go right/left/center, and 0 means 
continue doing what you are doing
roboGo actually sends input to the motors, depending on the state
*/
void loop(){
  USB.refresh();               //refresh USB.  
  state=checkSensors(state);   //get the current state, and see if it has changed since the last time.
  if(state){                   //if state is not 0, sent the state to robogo!
    roboGo(state);
  }
}

//State    |     Do
//  0      |   Check sensors again!
//  1      |   Go Straight
//  2      |   Go Right
//  3      |   Go Left



int checkSensors(int state){
  int delta=(analogRead(A0)-analogRead(A1));//Right sensor-Left sensor
  //send some data over usb
  USB.set(0,analogRead(A0));
  USB.set(1,analogRead(A1));
  USB.set(2,state);
  USB.set(3,delta);
  
  int newState;//temporary variable to hold the new state
  
  if(delta<0){//left sensor brighter than right, go left
    newState=3;
  }
  if(delta>0){//right sensor is brigher than left, go right
    newState=2;
  }
  if(delta==0){//if they are equal go straight
    newState=1;
  }
  
  if(newState!=state){//if the state has changed, send back the new state
    return newState;
  }
  else{
    return 0;//otherwise return 0
  }
    
    
}

void roboGo(int state){  //change output to motors and to status LEDs.
  switch (state){
    case 1:
    digitalWrite(9,HIGH);
    digitalWrite(10,HIGH);
    digitalWrite(7,HIGH);
    digitalWrite(8,HIGH);
    break;
    
    case 2:
    digitalWrite(9, LOW);
    digitalWrite(10,HIGH);
    digitalWrite(7,LOW);
    digitalWrite(8,HIGH);
    break;
    
    case 3:
    digitalWrite(9, HIGH);
    digitalWrite(10,LOW);



    digitalWrite(7,HIGH);
    digitalWrite(8,LOW);
    break;
  }
  }
    

Circuit for driving motors:

The capacitor should have a capacitance of 10uf.  The transistors are 2n3904s, and the battery pack is a 
2-cell AA pack, supplying 3 volts.

The LDRS on pins A0 and A1 should be grounded on one side, and pull up resistors should be on.

Exercises:

1. Make the motor output proportional to the LDR input on pins A0 and A1 instead of just on/off. 
What does this do to the path of the robot?

2. Make the robot avoid light!
3. Make the robot head search out an area of medium lighting (avoid areas that are too dark or too 

light)


